
Das bündeln von Geschäftsregeln in sogenannten Geschäftsobjekten vereinfacht nicht nur die Wiederverwendung der programmierten Regeln sondern auch den Aufruf aus anderen Anwendungen heraus via COM. Dieser Abschnitt führt Sie in die Grundlagen der COM-Server-Erstellung mitsamt der Fehlerbehandlung sowie der Integration sowohl in den Microsoft Transaktion Server als auch in die VFP Komponentengalerie ein.


Erstellen und Verwalten von COM-Komponenten

Microsofts Component Object Model (COM) ist eine der Technologien, die einfach Sinn machen. Sie ist Voraussetzung für objektorientierte Entwicklung und ein mehrschichtiges Design. Durch COM können Sie Ihre Objekte auch für andere Software und Programmierumgebungen verfügbar machen. Daher handelt es sich hier um eine flexible Technologie für Geschäftskomponenten der mittleren Schicht. Wir zeigen Ihnen, wie Sie COM-Komponenten erstellen, die unterschiedlichen Arten der COM-Komponenten, wie Sie diese registrieren und wie sie mit Hilfe der Komponenten-Gallerie verwaltet werden. Außerdem gehen wir auf die Fehlerbehandlung mit COM ein sowie auf spezielle Problematiken beim Einsatz von COM mit dem Microsoft Transaction Server.

Evan ist ein Fan der in der BBC gezeigten Serie „Junkyard Wars“. Zwei Teams, die aus je vier Personen bestehen, müssen je ein Gerät bauen, das eine bestimmte Aufgabe ausführt. Dafür stehen ihnen zehn Stunden zur Verfügung und sie dürfen nur Gegenstände benutzen, die sie von einem Schrottplatz ergattern. Die Teams haben nur die einfachsten Informationen über die Geräte, die sie zusammensetzen sollen. Der Showmaster stellt eine Aufgabe wie „bauen Sie eine Flugmaschine“ oder „erstellen Sie ein Gerät, mit dem sich ein untergegangenes Auto bergen lässt“. In Evans Lieblingsfolge mussten die Teams ein Gerät bauen, das eine Mauer einreißt. Ein Team erstellte einen Rahmen in Form eines großen A, an dem an einer Kette hängend eine Stange aufgehängt war. An einem Ende der Stange war ein schweres Gewicht angebracht und das Team bewegte die Stange mit Muskelkraft vor und zurück. Das zweite Team erstellte ein hydraulisches Gerät mit einer Klaue an einem Ende. Um es kurz zu machen: das Team mit der Hydraulik verlor, da ihr kompliziertes System kaputt ging und weil die Mauern ein Dach hatten, das sie nicht greifen konnten. Das Team mit der Stange war sehr langsam, aber da die Konstruktion sehr einfach war, hielt sie. Die Stange war außerdem vielseitig; sie konnten die Kette verlängern und verkürzen, so dass sie die Mauer an jedem Punkt einreißen konnten.

Mit wenigen Informationen gewinnen die Teams diese Wettbewerbe mit einfachen und flexiblen Konstruktionen. Die Einfachheit gibt dem Designer die Freiheit, sich auf das Problem zu konzentrieren statt die Lösung zu erstellen. Durch die Vielseitigkeit können auch unerwartete Situationen mit geringen Änderungen des Systems gemeistert werden. Das Component Object Model (COM) und Distributed COM (DCOM) unterstützen Visual FoxPro-Entwickler sowohl mit Einfachheit als auch mit Vielseitigkeit bei der Erstellung der Anwendungen. Sehen wir uns an, durch welche Eigenschaften sie uns unterstützen, unsere Anwendungen einfach und vielseitig zu gestalten. 

Was ist COM?

COM ist eine Technologie, die es einem Entwickler ermöglicht, Objekte, die in einer Sprache geschrieben wurden, in einer großen Anzahl anderer Programmiersprachen oder anderer Software, beispielsweise Microsoft Office, einzusetzen. COM ermöglicht es uns, eine Anwendung auf einem einzelnen Rechner oder via DCOM auf unterschiedlichen Servern zu installieren.

Die Einfachheit entsteht aus der Art, in der COM implementiert wurde. Nehmen Sie eine in Visual FoxPro entwickelte Klassendefinition, fügen Sie ihr das Schlüsselwort OLEPUBLIC hinzu und kompilieren Sie sie in eine .EXE oder .DLL. Anschließend können Sie die Klasse in jeder der folgenden Arten instanziieren:

VFP:

oJoe = CREATEOBJECT(“Staff“)

COM:

oJoe = CREATEOBJECT (“ComName.Staff“)

DCOM:

oJoe = CREATEOBJECT(“DcomName.Staff“,;






“MyServer.com”)

Nachdem ein Objekt instanziiert ist, wird es immer auf die gleiche Art verwendet, unabhängig davon, ob VFP, COM oder DCOM eingesetzt wird. Was könnte noch einfacher sein?

COM bietet uns bei der Verteilung eine große Anzahl an Wahlmöglichkeiten. Wir können Anwendungen lokal auf einer einzelnen Maschine laufen lassen. Wenn notwendig können wir zu DCOM umschalten, wenn die Anwendung von mehreren Anwender genutzt und auf einem Server ausgeführt wird.

Da COM für die Implementierung von Objekten eingesetzt wird, ist es ein natürlicher Teil der objektorientierten Programmierung (OOP) in Visual FoxPro.

Erstellen von COM-Komponenten

Da COM-Komponenten lokal eingesetzt und anschließend auf remote Server skaliert werden können, sind sie ideal für die Kapselung von Geschäftsregeln in der mittleren Schicht. So lange jede Software nicht direkt, sondern über die COM-Komponenten auf die Daten zugreift, können Sie sicher sein, dass Ihre Geschäftsregeln beachtet werden. Wie Sie in Listing 1 sehen können, sind die Geschäftsregeln einfach zu erstellen und da sie in der Komponente zentral gehalten werden, sind sie auch einfach zu ändern.

A) Die Arten der COM-Server

COM-Komponenten können in-process oder out-of-process erstellt werden. In-process-Komponenten können single-threaded oder multi-threaded erstellt werden. Während des Kompilierens entscheiden Sie, welche Art der Komponente Sie erstellen wollen: out-of-process .EXE, single-threaded In-Process .DLL oder multi-threaded in-process .DLL.

Eine In-process-Komponente ist eine .DLL, die im gleichen Adressbereich ausgeführt wird wie das Programm, das sie aufruft. In-process-Komponenten laufen schneller, da keine Kommunikation zwischen unterschiedlichen Prozessen ausgeführt werden muss. Der Nachteil besteht darin, dass ein Absturz der .DLL auch den aufrufenden Client mit runterzieht. Um eine neue Version einer .DLL zu installieren, die Teil eines Webservers ist, müssen Sie den Webserver herunterfahren. Die einzige Möglichkeit, eine In-process-Komponente auf einem remoten Server auszuführen ist innerhalb des MTS. Eine multi-threading .DLL kann die Antwortzeiten des Systems verbessern, besonders wenn langwierige Prozesse ausgeführt werden. Ein kürzerer Prozess ist in der Lage, den langwierigen Prozess zu unterbrechen und vermittelt den Anwendern den Eindruck einer kürzeren Antwortzeit. Multi-threading geht auf Kosten des Overhead für die Verwaltung der Prozesse.

Eine Out-of-process-Komponente ist eine .EXE, die in einem separaten Prozess getrennt vom aufrufenden Client ausgeführt wird. Obwohl sie langsamer ausgeführt wird als eine .DLL (wie bereits oben bemerkt), zieht ein Absturz auf dem Server den Client nicht mit. Eine .EXE kann ausgetauscht werden, ohne dass der Webserver heruntergefahren werden muss. Die Out-of-process-Komponente kann via DCOM auf einem remoten Rechner ausgeführt werden.

B) Die Geschäftsregeln

Geschäftsregeln sind die Abläufe im System, die Sie modellieren. Beispiele für Geschäftsregeln sind:

· Jede Rechnung muss mindestens einen Artikel enthalten.

· Alle Patienten müssen mindestens 18 Jahre alt sein.

· Kunden, die mit Scheck bezahlen, müssen sich identifizieren.

Listing 1. Eine COM-Komponente zeigt uns die Geschäftsregeln für das Hinzufügen von Patienten.

DEFINE CLASS Patient AS SESSION ;








OLEPUBLIC


* Adds new patient to database


* Returns .T. if successful


* Returns .F. if a business

* rule is violated


* private datasession


DATASESSION = 2


PROCEDURE Init



USE Patients SHARED in 0


ENDPROC


PROCEDURE Destroy



USE IN Patients


ENDPROC


PROCEDURE AddData



LPARAMETERS pcName, pnAge



* Patient name must not be 



* blank (“required”)



IF ALLTRIM(pcName) == ""




RETURN .F.



ENDIF



* Patient must be 18 or over



IF pnAge < 18




RETURN .F.



ENDIF



* No business rule violated, 



* add data to database



SELECT Patients



INSERT INTO Patients ;






(FullName, Age) ;




VALUES (pcName, pnAge)



RETURN .T.


ENDPROC

ENDDEFINE

Anschließend wird die Komponente in eine COM-.DLL mit Namen BIZCOM.DLL kompiliert. Listing 2 zeigt Ihnen ein Beispiel für den Einsatz in VFP. Sie müssen dies alles nicht manuell eingeben. Alle Komponenten und Tabellen finden Sie in den Downloads zu diesem Buch unter www.hentzen werke.com. Kopieren Sie alle Komponenten und Tabellen in ein Verzeichnis und machen Sie es zum aktuellen Verzeichnis. Evan setzt statt custom-basierter Klassen session-basierte Klassen ein, da sie es ihm ermöglichen, mit privaten Datensitzungen zu arbeiten. Dies bedeutet für nicht-COM-basierte Anwendungen, dass die Datenmanipulationen innerhalb der Objekte sich nicht gegenseitig beeinflussen. Außerdem hat dies den Vorteil, dass Sie diese Klassendefinitionen sowohl in einer reinen VFP-Umgebung als auch in einer COM-Umgebung nutzen können.

Listing 2 benutzt eine COM-Komponente mit gültigen und ungültigen Daten.

* Instantiate the CLASS as an object

oPatient = CREATEOBJECT("BizCom.Patient")

* Add a patient

IF oPatient.AddData("Humphrey Bogart";


, 82) = .T.


MESSAGEBOX("Data saved")

ELSE


MESSAGEBOX("Data not saved")

ENDIF

* Attempt to add another patient that

* breaks the rules

IF oPatient.AddData("Orphan Annie",;


5) = .T.


MESSAGEBOX("Data saved")

ELSE


MESSAGEBOX("Data not saved")

ENDIF

*Release the object

RELEASE oPatient

C) Zusammenarbeit mit anderen Komponenten

COM-Komponenten können sich gegenseitig aufrufen, um Aufgaben an andere Komponenten zu delegieren. In Listing 3 delegiert das Objekt oEmployee eine Suche in einer Tabelle an das Objekt oDept. Listing 4 zeigt Ihnen wie die Methode im Code aufgerufen wird. Den Code von oDept haben wir hier nicht veröffentlicht, da wir nicht sicher sind, wie diese Aufgabe implementiert ist. Alles was wir wissen ist, dass wir dem Objekt einen Primärschlüssel übergeben und dass wir den Namen zurückbekommen. Diese Daten kann das Objekt aus einer VFP-Tabelle auf C:\ oder auch aus einer Tabelle des SQL Server am anderen Ende der Welt erhalten.

Listing 3. Das Objekt oEmployee delegiert eine Tabellenabfrage an das Objekt oDept.

DEFINE CLASS Employee AS SESSION ;








OLEPUBLIC


* Employee Biz object


* private datasession


DATASESSION = 2 


PROCEDURE Init



use Employee SHARED IN 0


ENDPROC


PROCEDURE Destroy



USE IN Employee


ENDPROC


PROCEDURE GetDeptName



LPARAMETERS pnEmployee PK



LOCAL lcDeptName



* Get the department foreign 



* key from employee table



* Returns department name 



* if found



IF NOT USED("Employee")




USE Employee in 0



ENDIF



IF SEEK(pnEmployeePK, ;




"Employee", "pkId")




* Delegate retrieving the




* department name to 




* oDept object



oDept = CREATEOBJECT("Dept")



lcDeptName = oDept.GetName(;





     Employee.fkDept)




RELEASE oDept




RETURN lcDeptName



ELSE




RETURN "Not found"



ENDIF


ENDPROC

ENDDEFINE

Listing 4. Verwenden der Methode GetDeptName() auf das Objekt oEmployee

* Instantiate employee business class

oEmployee = ;


CREATEOBJECT("BizCom.Employee")

* Get the dept manes for employees 

* with primary keys 1, 2, 4, 5

? OEmployee.GetDeptName(1)

? OEmployee.GetDeptName(2)

? OEmployee.GetDeptName(4)

? OEmployee.GetDeptName(5)

* Release the employee object 

* from memory

RELEASE oEmployee

D) Fehlerbehandlung mit COM

Die Fehlerbehandlung ist bei COM-Komponenten besonders wichtig, da diese Komponenten häufig über keine Benutzerschnittstelle verfügen und sie auf Rechnern ohne Anwender ausgeführt werden. Wenn in einer konventionellen Anwendung Ihr Programm im Falle eines Fehlers nicht entscheiden kann, wie sinnvoll weitergearbeitet wird, können Sie dem Anwender die Möglichkeiten anzeigen. Dies geschieht meist klassisch: „Auf die Daten kann nicht zugegriffen werden. Wollen Sie es erneut versuchen, ignorieren oder abbrechen?“ Der Anwender kann dann versuchen, mit einer oder mehreren dieser Möglichkeiten die Anwendung wieder zum Laufen zu bringen, bevor er den Support anruft oder den Rechner neu bootet. Eine COM-Komponente kann lediglich der aufrufenden Komponente den Fehler melden und ihn in eine Logdatei schreiben. Würde durch eine COM-.EXE eine Meldung ausgegeben, könnte dadurch die Komponente blockiert werden, da die Meldung auf dem Monitor des Servers angezeigt wird. Wenn dort niemand anwesend ist, der auf die Schaltfläche OK klickt, wartet die Komponente auf eine Eingabe und kann in dieser Zeit auf keine Anfragen reagieren. Sie können diese Wartesituation (beispielsweise durch die Fehlermeldung eines nicht behandelten Fehlers) durch den Einsatz von SYS(2335) verhindern. COM-.DLLs haben keine Benutzeroberfläche, so dass SYS(2335) überflüssig ist.

Eine gute Strategie der Fehlerbehandlung ahnt die Probleme bereits im Vorfeld und behandelt sie  elegant. Es ist immer sinnvoll zu prüfen, ob alle Parameter übergeben werden und ob sie auch den richtigen Typ haben. Statt einfach eine Tabelle zu öffnen, prüfen Sie mit der Funktion USED(), ob diese vielleicht schon offen ist. Wenn Sie die Ergebnisse prüfen, die Sie von einer Funktion oder einem Objekt erhalten haben, testen Sie auf alle Rückgabewerte, auch auf „unmögliche“. Auch wenn wir hier die Fehlerbehandlung unter COM diskutieren, gelten diese Ausführungen natürlich für alle Arten der Anwendungen.

Bei nicht vorhersehbaren Fehlern ist es ein gutes Vorgehen, so viele Informationen wie möglich zurückzugeben. Jede Klassendefinition sollte einen Code in ihrer Methode Error() haben, der beim Auftreten eines Fehlers automatisch ausgeführt wird. Dieses Vorgehen unterscheidet sich von dem in konventionellen Anwendungen, wo Sie eine Fehlerbehandlung mit dem Befehl ON ERROR aufrufen. Wenn das Fehlerereignis eines Objekts ausgelöst wird, sollte die Methode Error() versuchen alle Fehler abzuhandeln, die für diese Klasse spezifisch sind. Fehler, die nicht behandelt werden können, können in der Klassenhierarchie hochgereicht werden. Kann auch die übergeordnete Klasse den Fehler nicht behandeln, kann eine spezielle Klasse für die Behandlung von Fehlern aufgerufen werden. 

Mit Hilfe der Funktion COMRETURN ERROR() können COM-Komponenten Fehler in der Hierarchie hochreichen. Das Schöne an diesem Mechanismus ist, dass die Clientschicht COM-Fehler nicht anders behandeln muss als Fehler innerhalb der Clientschicht selber. Die Fehlerinformation erhalten Sie über AERROR(). Listing 5 zeigt Ihnen den Aufruf von COMRETURN ERROR() und demonstriert eine Möglichkeit, den Fehler in einer Datei festzuhalten. Wir haben versucht, ein einfaches Beispiel zu zeigen; in einer echten Komponente würden Sie zunächst versuchen, den Fehler lokal zu behandeln.

Listing 5. Fehlerbehandlung und Logging in einer COM-Komponente

DEFINE CLASS Invoice AS SESSION OLEPUBLIC


* Invoice Biz object


* Demonstrates COM error


* handling


PROCEDURE DoSomething(pnDeptPK)



* generate a fake error 



* using the error command


ERROR "A realy terrible error"


ENDPROC


PROCEDURE Error(pnErrorNum, ;




pcMethodName, pnLineNum)



LOCAL lcErrorMess



* To keep example simple, 



* don't handle error, pass 



* it back to calling 



* application 



lcErrorMess = ;




DTOC(DATETIME())+ ;




" | Error Number:" + ;




TRANSFORM(pnErrorNum) + ;




" | Method Name:" + ;




pcMethodName + ;




" | Line Number:" + ;




TRANSFORM(pnLineNum) + ;




" | Message:" + message()



* Append to error log file



STRTOFILE(lcErrorMess + ;



chr(13), "error.log", .t.)



* Raise a COM error to



* notify client



* The Client will handle the 



* error as instructed



COMRETURNERROR("MyServer", ;







lcErrorMess)


ENDPROC

ENDDEFINE

Listing 6 zeigt, dass im Code des Client COM-Fehler exakt wie andere VFP-Fehler behandelt werden. Wir zeigen hier nur eine Information an, dass ein Fehler aufgetreten ist. In einer echten Anwendung würden wir den Fehler eleganter behandeln.

Listing 6. Behandlung eines COM-Fehlers durch den Client.

* Enable error handler

ON ERROR ErrorHandler()


* Instantiate invoice class to 

* an object

oInvoice = CREATEOBJECT(; 





"BizCom.Invoice")

nResult = oInvoice.DoSomething(1)

PROCEDURE ErrorHandler()


DIMENSION laError[1]


* Load error info into an array


* In a real application we 


* would try to handle the error


AERROR(laError)


FOR i = 1 TO 7



LcErrorMessage = ;





lcErrorMessage + ;





laError(i) + CHR(13)


ENDFOR


MESSAGEBOX(lcErrorMessage)

ENDPROC

E) Entwerfen von Komponenten für den MTS

Der Microsoft Transaction Server (MTS) ist eine Laufzeitumgebung für COM-Komponenten, die das Caching der Komponenten für eine verbesserte Performance ermöglicht. Worum es sich beim MTS handelt und weshalb Sie ihn einsetzen sollten, lesen Sie im Artikel „Microsoft Transaction Server für Visual FoxPro Developers“ von Randy Brown, der im Abschnitt „Weiterführende Literatur“ am Ende dieses Kapitels aufgelistet ist. 

Beim Entwerfen von COM-Komponenten, die innerhalb des MTS ausgeführt werden sollen, müssen wir die folgenden Punkte beachten:

· Statuslose Komponenten, die nicht auf einem gespeicherten Status basieren, sind innerhalb des MTS effizienter.

· COM-Komponenten, die innerhalb des MTS ausgeführt werden sollen, müssen immer als In-Prozess-.DLLs kompiliert werden.

· Die Methoden SetComplete und SetAbort teilen dem MTS mit, dass Sie die Komponente eine Zeitlang nicht mehr benötigen.

Statuslose Komponenten sind im MTS effizienter, da der Kontext der Komponente nicht bei jedem Aufruf neu wiederhergestellt werden muss. Das bedeutet weniger Arbeit für den Server, der daher mehr Anfragen behandeln kann und daher besser skalierbar ist. Wie unser technischer Herausgeber bemerkte, sind alle Codebeispiele, die wir bislang in diesem Kapitel benutzt haben, statusbehaftet, da sie auf den Tabellen basieren, die in der Methode Init() des Objekts geöffnet werden. Nähere Informationen über statuslose und statusbehaftete Objekte finden Sie unter http://msdn.microsoft.com/ library/techart/d3dalchg.htm.

Im MTS können nur In-process-.DLLs ausgeführt werden. Da Ihre Out-of-process-.EXEs nicht innerhalb des MTS laufen, müssen Sie sie neu als .DLLs kompilieren.

Auch wenn wir keine Transaktionen einsetzen, benutzen wir SetComplete und SetAbort, damit das von uns benutzte Objekt seinen Status verliert. Dann kann der MTS das Objekt deaktivieren und Ressourcen auf dem Server freigeben. Setzen Sie am Ende jeder Methode des Objekts vor der Anweisung RETURN ein SetComplete ab.

F) Transaktionsunterstützung für COM-Komponenten im MTS

Der MTS unterstützt Datenbanktransaktionen über den Distributed Transaction Coordinator (DTC). Sowohl der Microsoft SQL Server als auch Oracle werden unterstützt, leider aber die Datenbanktransaktionen von Visual FoxPro nicht. Wenn ein Objekt mit den Tabellen von Visual FoxPro arbeitet, muss es die Transaktionen selbst verwalten. Der DTC geht davon aus, dass Transaktionen immer vom MTS behandelt werden und dass Transaktionen entweder abgeschlossen oder über heterogene Systemdatenbanken zurückgenommen werden können. Vor dem Erscheinen des DTC musste der Entwickler diese potentiell sehr komplexen Transaktionen selbst verwalten.

Um eine Transaktion zu beginnen, rufen Sie GetObjectContext(.T.) auf. Anschließend führen Sie die Datenbankoperationen aus. Am Schluss rufen Sie SetComplete() auf, um die Transaktion abzuschließen oder SetAbort(), um sie zurückzunehmen.

Verwalten von COM-Komponenten

Nun haben wir uns über die Entwicklung von COM-Komponenten unterhalten; lassen Sie uns nun hinter die Szenerie des Erstellens und Verwaltens der Komponenten blicken. Wir beschäftigen uns mit der Komponenten-Galerie für die Verwaltung der Komponenten, den einzelnen COM-Servertypen, der Registrierung der Komponenten sowie mit der Erstellung und Verwaltung der Packages mit dem MTS Explorer.

G) Benutzen der Komponenten-Galerie

Die Komponenten-Galerie rufen Sie im Menü Extras -> Komponenten-Gallery auf. Die Galerie ist eine Oberfläche für die Gruppierung von Komponenten in Kategorien, „Kataloge“ genannt. Kataloge sind Zeiger oder Shortcuts auf die aktuellen Komponenten und können in Ordner unterteilt werden. Damit bieten sie uns einen bequemen Weg nicht nur Komponenten zu organisieren, sondern auch andere Klassen, Projekte und eigentlich alles andere. Jeder Katalog ist eine .DBF-Datei. Die Komponenten-Galerie ist ein weiteres Werkzeug, das uns hilft die Entwicklung mit VFP zu organisieren. Sie ist projektübergreifend und geht weiter als lediglich zu den .VCX-basierten Klassen wie der Klassenkatalog. 

So legen Sie einen neuen Katalog an:

1. Klicken Sie auf die Schaltfläche „Optionen“ in der Toolbar.

2. Wechseln Sie auf den Tab „Kataloge“ (vgl. Abbildung 1)

3. Klicken Sie auf die Schaltfläche „Neu...“

4. Geben Sie einen Namen und den Ort ein, an dem der Katalog angelegt werden soll.

5. Klicken Sie auf „Speichern“.

6. Klicken Sie auf „OK“, um die Optionen zu verlassen.

So legen Sie einen neuen Ordner an:

1. Wählen Sie im linken Fenster den Katalog, dem Sie einen Ordner hinzufügen wollen.

2. Klicken Sie mit der rechten Maustaste in das rechte Fenster der Komponenten-Galerie.

3. Wählen Sie „Neu“, anschließend „Ordner“.

4. Klicken Sie mit der rechten Maustaste auf den Ordner „Neuer Ordner“ und anschließend „Eigenschaften“.

5. Geben Sie einen Namen und eine Beschreibung ein und klicken dann auf „OK“.

So fügen Sie einem Katalog eine Komponente hinzu:

1. Wählen Sie den Katalog oder Ordner, dem die Komponente hinzugefügt werden soll.

2. Klicken Sie mit der rechten Maustaste ins rechte Fenster.

3. Wählen Sie „Neu“, dann den Typ der Komponenten (ActiveX, Klasse, Datenquelle, Datei, Formular, Abbildung, Menü, Programm, Projekt, Bericht, Beispiel, Website).

4. Durchsuchen Sie nach der Komponente und klicken Sie dann auf „OK“.

5. Klicken Sie mit der rechten Maustaste auf die Komponente und wählen Sie „Eigenschaften“, um ihr eine Beschreibung und ein Icon hinzuzufügen.

Sie können innerhalb der Komponenten-Galerie unterschiedliche Optionen einstellen, indem Sie auf die Schaltfläche „Optionen“ in der Toolbar klicken. Ein sehr brauchbares Feature ist die Fähigkeit, Komponenten einfach in Ihre VFP-Projekte zu ziehen.

Abbildung 1.Der Standard-Tab der Komponenten-Galerie. Wenn Sie auf eine Komponente geklickt haben, können Sie das Standardverhalten wählen.

Abbildung 2. Der Tab "Optionen" der Komponenten-Galerie. Evan hat seine eigenen Kataloge hinzugefügt
.

H) Registrierung und Rückgängigmachen der Registrierung von COM-Komponenten

COM-Komponenten müssen unter Windows registriert werden (also in der Registry), bevor Sie sie nutzen können. Für .EXE-Komponenten:

· Registrieren durch Ausführen der Anwendung mit MyExe /regserver

· Rückgängigmachen der Registrierung durch MyExe /unregister

Für .DLL-Komponenten:

· Registrieren mit REGSVR32  und Parameter MyDll.DLL

· Rückgängigmachen der Registrierung durch REGSVR32 /u MyDll.DLL

I) Erstellen von Packages und Verwalten der Komponenten mit dem MTS Explorer

Im MTS können Sie sich die Komponenten ansehen, die auf einer Maschine registriert sind. In Kapitel 10, „Deploying an Application“ finden Sie nähere Details dazu. Vergleichen Sie auch das White Paper von Microsoft: „Microsoft Management Console: Overview“, das Sie am Ende dieses Kapitels im Abschnitt „Weiterführende Literatur“ finden. 

Im Folgenden führen wir einige allgemeine Aufgaben auf, die mit dem MTS Explorer (oder der Microsoft Management Console) ausgeführt werden. Abbildung 3 zeigt das Layout der Microsoft Management Console, eine Oberfläche, die allen vertraut ist, die mit dem Windows Explorer gearbeitet haben.

Abbildung 3. Das Package SiBiz in der Microsoft Management Console

Ein leeres Package erstellen:

1. Führen Sie MMC (Microsoft Management Console) von der Befehlszeile aus.

2. Navigieren Sie im Baum des linken Fensters zu den „Packages Installed“ (vgl. Abbildung 3).

3. Klicken Sie in der Toolbar auf die Schaltfläche „Erstellt ein neues Objekt“.

4. Wählen Sie „Ein leeres Paket erstellen“.

5. Geben Sie einen Namen für das Package an und klicken Sie auf „Weiter“.

6. Wählen Sie die Anwender, die das Package nutzen dürfen und klicken Sie auf „Fertigstellen“.

Um einem Package eine Komponente hinzuzufügen:

7. Markieren Sie im linken Fenster das Package.

8. Wählen Sie unter dem Package den Ordner „Komponenten“.

9. Benutzen Sie den Windows Explorer um Ihre Komponente zu finden.

10. Ziehen Sie sie in das rechte Fenster.

J) Einsetzen der rollenbasierten Sicherheit

Der MTS ermöglicht es Ihnen, die Sicherheitsfeatures von Windows NT und Windows 2000 mit Hilfe der rollenbasierten Sicherheit in ihrer Anwendung zu nutzen. Eine Rolle ist eine Anwendergruppe, die auf eine MTS-Komponente zugreifen kann.

Um einem Package eine Rolle hinzuzufügen:

1. Markieren Sie den Unterordner Rollen unterhalb dem Package im linken Fenster.

2. Klicken Sie mit der rechten Maustaste auf den Unterordner und wählen Sie Neu, dann Rollen.

3. Geben Sie einen Namen für die Rolle ein und klicken Sie auf OK.

4. Markieren Sie im linken Fenster Ihre Rolle und dort den Unterordner Benutzer.

5. Klicken Sie mit der rechten Maustaste darauf und wählen Sie neu, dann Benutzer.

6. Wählen Sie alle Anwender, die Sie der Rolle zuordnen wollen und klicken auf OK.

Beispielfrage

Sie wollen eine Komponente mit dem Namen TheDLL.DLL registrieren. Welches ist die richtige Syntax?
A. TheDLL.DLL /R

B. TheDLL.DLL /Reg

C. REGSVR32 /R TheDLL.DLL

D. REGSVR32 TheDLL.DLL

Antwort: D

K) Weiterführende Literatur

· Internet Applications with Visual FoxPro 6.0, Rick Strahl, Kapitel 9 “Visual FoxPro and COM”

· “Error Handling in Visual FoxPro” Doug Hennig, enthält Artikel und Code für die Behandlung von Fehlern während der Ausführung, www.stonefield.com/pub/errorh.zip
· http://fox.wikis.com/wc.dll?Wiki~ComErrorHandling
· http://fox.wikis.com/wc.dll?Wiki~COMComponentExample
· http://fox.wikis.com/wc.dll?Wiki~ComReturnError
· http://fox.wikis.com/wc.dll?Wiki~ComErrors
· http://fox.wikis.com/wc.dll?Wiki~ErrorEventStrategy
· „Microsoft Management Console: Overview“ White Paper, http://microsoft.com/windows2000/library/howitworks/management/mmcover.asp
· “Microsoft Transaction Server for Visual FoxPro Developers” Randy Brown, Microsoft Corporation, http://msdn.microsoft.com/library/default.asp?URL=/library/techart/ntsvfp.htm
· “Using Microsoft Transaction Server With VFP”, Rick Strahl, www.westwind.com/presentations/mts/mts.htm
· http://fox.wikis.com/wc.dll?Wiki~SettingUpMTS
· “How Microsoft Transaction Server Changes the COM Programming Model”, David Chappell, http://msdn.microsoft.com/library/periodic/period98/mtsjan.htm
· http://fox.wikis.com/wc.dll?Wiki~UsingAndDevelopingForMTS
· “The Microsoft Component Gallery”, Steven M. Black, http://msdn.microsoft.com/library/default.asp?URL=/library/techart/Vfpgallery.htm
· Advanced Object Oriented Programming with Visual FoxPro 6.0, Markus Egger, Kapitel 5, “OOP Standards”

· The Fundamentals: Building Visual Studio Applications on a Visual FoxPro 6.0 Foundation, Whil Hentzen, Kapitel 17, “The Component Gallery”

